# $Id: PKGBUILD 174021 2013-01-01 15:15:16Z foutrelis $
# Maintainer: Allan McRae <allan@archlinux.org>
# Contributor: judd <jvinet@zeroflux.org>

pkgname=patch
pkgver=2.7.1
pkgrel=2
pkgdesc="A utility to apply patch files to original sources"
arch=('i686' 'x86_64')
url="http://www.gnu.org/software/patch/patch.html"
license=('GPL')
groups=('base-devel')
depends=('glibc')
makedepends=('ed')
optdepends=('ed: for patch -e functionality')
source=(ftp://ftp.gnu.org/gnu/${pkgname}/${pkgname}-${pkgver}.tar.xz{,.sig}
        patch-2.7.1-initialize-data-structures-early-enough.patch)
md5sums=('e9ae5393426d3ad783a300a338c09b72'
         'b12189e0de3cb2af25268441647ec517'
         'dc6367a7cd49933d4006c246789e98da')

build() {
  cd ${srcdir}/${pkgname}-${pkgver}

  # Fix segfault on non-numeric strip-count
  # (also segfaults on nonexistent directory passed to -d)
  # http://savannah.gnu.org/bugs/?37500
  patch -Np1 -i \
    $srcdir/patch-2.7.1-initialize-data-structures-early-enough.patch

  ./configure --prefix=/usr
  make
}

check() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make check
}

package() {
  cd ${srcdir}/${pkgname}-${pkgver}
  make DESTDIR=$pkgdir install
}
